





KidLisp Language Reference

@jeffrey

Aesthetic.Computer

ORCID: 0009-0007-4460-4913

working draft — not for citation

March 2026 · b3a48ec6a

Dansk · Español · 中文 · 日本語

Index

1. Syntax	6
2. The 12 Categories	7
2.1 Graphics (9 functions)	7
2.2 Transformations (11 functions)	8
2.3 Math (14 functions)	8
2.4 Color (19+ named colors)	9
2.5 System & Display (9 functions)	10
2.6 Control Flow (7 functions)	10
2.7 Timing	11
2.8 Text (4 functions)	12
2.9 Input (3 functions)	12
2.10 Audio (6 functions)	13
2.11 3D Graphics (8 functions)	13
2.12 Composition via \$-Codes	14

3. Evaluation Model	16
3.1 Deterministic Randomness	16
3.2 Chaos Mode	17
4. Design by Omission	18
5. Complete Built-in Index	20
6. Example: Complete Program	21

1. Syntax

KidLisp uses S-expression syntax [McCarthy, 1960]. Every expression is either an atom (number, string, symbol) or a list enclosed in parentheses:

```
; Atoms  
42 ; number  
"hello" ; string  
red ; symbol (color name)  
  
; Lists (function application)  
(wipe "navy") ; clear screen navy  
(ink "red") ; set drawing color  
(box 10 10 50 50) ; draw rectangle  
(circle 64 64 30) ; draw circle
```

There are no operators, no infix notation, no special syntax beyond parentheses, strings, numbers, and symbols. Arithmetic uses prefix notation:

```
(+ 1 2) ; => 3  
(* 3 (+ 1 2)) ; => 9  
(- 10 3) ; => 7  
(/ 100 4) ; => 25  
(% 17 5) ; => 2 (modulo)
```

2. The 12 Categories

2.1 Graphics (9 functions)

Drawing primitives operate in immediate mode—each call renders directly to the framebuffer.

```
(wipe "black") ; clear canvas
(ink "cyan") ; set color
(line 0 0 128 128) ; diagonal line
(box 10 10 40 40) ; rectangle
(circle 64 64 20) ; circle at center
(tri 32 10 10 54 54 54) ; triangle
(plot 64 64) ; single pixel
(flood 64 64) ; flood fill
(shape 10 10 50 10 30 50) ; polygon
```

fill and outline control whether shapes are solid or stroked (Processing-compatible):

```
(fill) ; filled shapes
(ink "red")
(circle 64 64 30) ; solid red circle

(outline) ; outlined shapes
(ink "white")
(circle 64 64 30) ; white ring
```

2.2 Transformations (11 functions)

Pixel-level transformations operate on the frame-buffer after drawing:

```
(scroll 1 0) ; shift pixels right
(zoom 1.01) ; slow zoom in
(spin 1) ; rotate 1 degree
(blur 2) ; gaussian blur
(contrast 1.2) ; increase contrast
(suck 0.5) ; radial displacement
(sort) ; sort pixels by brightness
```

Transformations *accumulate* across frames, producing animation without explicit state management:

```
(wipe "black")
(ink "white")
(circle 64 64 10) ; draw circle
(spin 2) ; rotate 2 deg/frame
(zoom 1.005) ; grow slightly/frame
; Result: spiraling, growing circle
```

2.3 Math (14 functions)

```
(sin (* frame 0.05)) ; sine wave
(cos (* frame 0.03)) ; cosine wave
(random) ; 0-255
```



```
(random 10) ; 0-9
(random 5 15) ; 5-14
(wiggle 10) ; random +/-5
(floor 3.7) ; => 3
(abs -5) ; => 5
(sqrt 16) ; => 4
(min 3 7 1) ; => 1
(max 3 7 1) ; => 7
```

random uses a seeded PRNG initialized from the program's short code, making all generative output **deterministic**: the same program always produces the same visual.

2.4 Color (19+ named colors)

Colors can be specified as names, RGB values, or gradient strings:

```
(ink "red") ; named color
(ink 255 0 0) ; RGB
(ink 255 0 0 128) ; RGBA (semi-transparent)
(ink "fade:red-blue") ; horizontal gradient
(ink (fade "cyan" "magenta" "vertical"))
(ink "rainbow") ; repeating rainbow
(ink "zebra") ; striped pattern
```

Available color names include all CSS named colors: red, green, blue, yellow, orange, purple, pink,

cyan, magenta, black, white, gray, brown, lime, navy, teal, gold, coral, salmon, crimson, lavender, indigo, violet, turquoise, tomato, and more.

A bare color name on the first line sets the background:

```
"navy"  
(ink "gold")  
(circle 64 64 30)  
; Navy background, gold circle
```

2.5 System & Display (9 functions)

```
(def w width) ; screen width  
(def h height) ; screen height  
frame ; current frame number  
time ; seconds elapsed  
  
(resolution "half") ; half-resolution  
(resolution 64 64) ; explicit 64x64  
(fps 30) ; set frame rate
```

2.6 Control Flow (7 functions)

```
; Variable definition  
(def x 50)  
(def y (+ 10 (* frame 0.5)))
```

```

; Conditional
(if (> frame 60)
  (ink "red")
  (ink "blue"))

; Loop
(repeat 10 i
  (circle (* i 12) 64 5))

; Execute once (not every frame)
(once (wipe "black"))

; Local binding
(let ((cx 64) (cy 64))
  (circle cx cy 30))

```

The `repeat` loop binds its counter variable (`i` above) from 0 to $n - 1$. Nested repeats use different variable names:

```

(repeat 8 i
  (repeat 8 j
    (box (* i 16) (* j 16) 14 14)))
; 8x8 grid of squares

```

2.7 Timing

KidLisp's timing system uses *inline scheduling* rather than callbacks. Timing expressions wrap

code blocks:

```
; Frame-based timing
(0 (wipe "black")) ; every frame
(1 (spin 5)) ; every 2nd frame
(30f (ink "red")) ; after 30 frames

; Second-based timing
(1s (wipe "red")) ; after 1 second
(2.5s (ink "gold")) ; after 2.5 seconds
(1s! (write "!" 10 10)) ; fire once at 1s

; Repeating intervals
(0.5s... (spin 45)) ; every 0.5 seconds
(1s... (scroll 1 0)) ; every second
```

This eliminates the need for `setTimeout`, `setInterval`, or animation frame callbacks. Time is a *property of the expression*, not a function called on it.

2.8 Text (4 functions)

```
(write "hello" 10 20) ; text at position
(write "centered" ; centered text
  (/ width 2) (/ height 2))
(type "typed" 10 10) ; typewriter effect
```

2.9 Input (3 functions)

```

; Pen/mouse position and state
(if pen
  (circle pen.x pen.y 5))

; Touch input
(if hand
  (line hand.x hand.y 64 64))

```

2.10 Audio (6 functions)

```

; Microphone input
(def vol (mic)) ; amplitude 0-255
(circle 64 64 vol) ; sound-reactive

; Melody playback
(melody "C4 E4 G4 C5")

; Overtone harmonics
(overtone 440 3) ; 3rd harmonic of A4

```

2.11 3D Graphics (8 functions)

```

(cube "mycube") ; create 3D cube
(form "mycube"
  (move 0 0 -5) ; position
  (scale 2) ; size
  (spin (* frame 0.5) 1 1 0)) ; rotate

```

2.12 Composition via \$-Codes

The most distinctive feature of KidLisp is its composition system. Instead of user-defined functions, programs reference other programs by their short code:

```
; Embed another program as a layer  
($cow) ; execute program "cow"  
($nece 10 20) ; pass parameters  
  
; Multi-layer composition  
(wipe "black")  
($cow) ; background layer  
($faim) ; foreground layer  
  
; Chain multiple programs  
(resolution 64 64)  
($zobi)  
(spin 1)  
($grel)
```

Programs are fetched from MongoDB, cached in RAM and IndexedDB, and content-addressed via SHA-256. The \$-code system replaces user-defined functions, modules, and libraries with a single social mechanism: *find a program someone made, reference it by code, compose with it.*

The most-referenced programs are simple visual primitives that serve as building blocks. The system produces a social dependency graph: program A embeds program B, which embeds program C, up to 7 layers deep in observed usage.

3. Evaluation Model

KidLisp re-evaluates its **entire AST every frame**. There is no retained state beyond explicitly defined variables. The frame and time variables increment automatically. Transformations like `zoom`, `scroll`, and `spin` accumulate in the framebuffer.

This model eliminates the `setup()/draw()` split familiar from Processing [Reas and Fry, 2007]. Every line of a KidLisp program executes every frame. Animation emerges from frame-dependent expressions:

```
(wipe "black")
(ink "white")
; Circle orbits center using sin/cos
(circle
  (+ 64 (* 30 (cos (* frame 0.03))))
  (+ 64 (* 30 (sin (* frame 0.03))))
  8)
```

3.1 Deterministic Randomness

`random` is seeded from the program's short code. The same program produces identical output on

every device, every time. This makes KidLisp programs *reproducible generative art*: the code *is* the artwork, not just the instructions for generating it.

3.2 Chaos Mode

Invalid input—gibberish, unbalanced parentheses, random characters—does not produce error messages. Instead, KidLisp detects low recognition ratios ($< 30\%$ known words) and high special character ratios ($> 50\%$) and renders artistic visual output. The user never sees “SyntaxError.” They see colors.

4. Design by Omission

KidLisp deliberately lacks:

- **User-defined functions:** replaced by \$-code composition
- **Recursion:** prevents infinite loops and stack overflows
- **File I/O:** the canvas is the only output
- **Networking:** programs cannot fetch external data
- **String manipulation:** no concat, split, replace
- **Error messages:** replaced by chaos mode
- **Import/require:** replaced by \$-codes
- **Objects/classes:** data is numbers, strings, and lists

Each omission is a design decision. The absence of user-defined functions forces composition through the social layer. The absence of error messages means beginners never encounter a wall of red text. The absence of networking means every program is self-contained and safe to run.

This follows the principle that a language's constraints define its creative space more than its features [Papert, 1980, Compton and Mateas, 2015]. KidLisp is not a general-purpose language that happens to do art. It is an art language that *cannot* do anything else.

5. Complete Built-in Index

Category	Count
Graphics	9
Transformations	11
Math	14
Color	19
System/Display	9
Control Flow	7
Text/Output	4
Input	3
Audio	6
3D Graphics	8
Composition	4
Data	4
Total	118

Table 1: All 118 built-in functions by category.

6. Example: Complete Program

A complete KidLisp program that draws a sound-reactive particle field with orbital motion:

```
"black"
(def vol (+ 5 (* (mic) 0.2)))
(repeat 50 i
  (def angle (* i 0.1256))
  (def r (+ 20 (* vol (sin (+ (* frame 0.02) angle)))))
  ↪ )
  (def cx (+ 64 (* r (cos (+ angle (* frame 0.01))))))
  (def cy (+ 64 (* r (sin (+ angle (* frame 0.01))))))
  (ink (+ 100 (* i 3)) 50 (+ 150 (* i 2)))
  (circle cx cy (+ 1 (random 3)))
(blur 1)
(zoom 1.002)
```

This program: sets a black background, reads microphone amplitude, draws 50 particles in orbital paths modulated by sound, colors them with a computed gradient, applies a slight blur, and slowly zooms in. It runs at 60 FPS. It is 12 lines long. It requires no imports, no setup function, no animation loop, and no event listeners.

ORCID: 0009-0007-4460-4913

References

Kate Compton and Michael Mateas. Casual creators. In *Proceedings of the International Conference on Computational Creativity*, 2015.

John McCarthy. Recursive functions of symbolic expressions and their computation by machine, part i. *Communications of the ACM*, 3(4):184–195, 1960.

Seymour Papert. *Mindstorms: Children, Computers, and Powerful Ideas*. Basic Books, 1980.

Casey Reas and Ben Fry. *Processing: A Programming Handbook for Visual Designers and Artists*. MIT Press, 2007.