



KidLisp sprogreference

118 indbyggede funktioner i 12 kategorier til generativ kunst

Syntaks, semantik, timing, farve og \$-kode kompositionssystemet

@jeffrey

Aesthetic.Computer

ORCID: 0009-0007-4460-4913

<https://aesthetic.computer>

[arbejdsudkast — ikke til citation]

Abstrakt. KidLisp er en minimal Lisp-dialekt til generativ kunst med 118 indbyggede funktioner fordelt på 12 kategorier, ingen brugerdefinerede funktioner, ingen rekursion, ingen fil-I/O og intet netværk. Det er sikkert ved konstruktion: et KidLisp-program kan ikke bryde ud af sin sandbox, crashe sin vært eller tilgå ressourcer ud over sit lærred. Denne artikel fungerer som en komplet sprogreference, der dokumenterer enhver indbygget funktion med syntaksfremhævede eksempler, timingsystemet (frame-baseret og sekundbaseret planlægning uden callbacks), farve-modellen (navngivne farver, RGB, gradienter, mønstre), \$-kode kompositionssystemet (indlejring af programmer ved reference som den primære abstraktionsmekanisme) og evalueringsmodellen (fuld AST-reevaluering hvert frame med deterministisk PRNG). Vi præsenterer KidLisp som et sprog designet omkring *udeladelse*: hvad det ikke kan, er lige så bevidst som hvad det kan, og begrænsningerne producerer et kreativt rum, som 59 forfattere har brugt til at skrive 16.634 programmer.

1. SYNTAKS

KidLisp bruger S-udtrykssyntaks [McCarthy, 1960]. Ethvert udtryk er enten et atom (tal, streng, symbol) eller en liste omsluttet af parenteser:

```
; Atomer
42                ; tal
"hello"           ; streng
red               ; symbol (farvenavn)

; Lister (funktionsapplikation)
(wipe "navy")      ; ryd skærm marineblå
(ink "red")        ; sæt tegnefarve
(box 10 10 50 50)  ; tegn rektangel
(circle 64 64 30)  ; tegn cirkel
```

Der er ingen operatorer, ingen infiks-notation, ingen speciel syntaks ud over parenteser, strenge, tal og symboler. Aritmetik bruger præfiks-notation:

```
(+ 1 2)           ; => 3
(* 3 (+ 1 2))     ; => 9
(- 10 3)          ; => 7
(/ 100 4)         ; => 25
(% 17 5)          ; => 2 (modulo)
```

2. DE 12 KATEGORIER

2.1 Grafik (9 funktioner)

Tegneprimitiver opererer i immediate-mode—hvert kald renderer direkte til framebufferen.

```
(wipe "black")      ; ryd lærred
(ink "cyan")        ; sæt farve
(line 0 0 128 128)  ; diagonal linje
(box 10 10 40 40)   ; rektangel
(circle 64 64 20)   ; cirkel i centrum
(tri 32 10 10 54 54 54) ; trekant
(plot 64 64)        ; enkelt pixel
(flood 64 64)       ; oversvømmelsesfyld
(shape 10 10 50 10 30 50) ; polygon
```

fill og outline kontrollerer, om former er udfyldte eller stregede:

```
(fill)              ; udfyldte former
(ink "red")         ; sæt farve
(circle 64 64 30)   ; massiv rød cirkel

(outline)           ; stregede former
(ink "white")       ; sæt farve
(circle 64 64 30)   ; hvid ring
```

2.2 Transformationer (11 funktioner)

Pixelniveau-transformationer opererer på frame-buffere efter tegning:

```
(scroll 1 0) ; flyt pixels til højre
(zoom 1.01) ; langsom zoom ind
(spin 1) ; roter 1 grad
(blur 2) ; gaussisk sløring
(contrast 1.2) ; øg kontrast
(suck 0.5) ; radial forskydning
(sort) ; sortér pixels efter
lysstyrke
```

Transformationer *akkumulerer* på tværs af frames, hvilket producerer animation uden eksPLICIT tilstandsstyring.

2.3 Matematik (14 funktioner)

```
(sin (* frame 0.05)) ; sinusbølge
(cos (* frame 0.03)) ; cosinusbølge
(random) ; 0-255
(random 10) ; 0-9
(random 5 15) ; 5-14
(wiggle 10) ; tilfældig +/-5
(floor 3.7) ; => 3
(abs -5) ; => 5
(sqrt 16) ; => 4
(min 3 7 1) ; => 1
(max 3 7 1) ; => 7
```

random bruger en seedet PRNG initialiseret fra programmets korte kode, hvilket gør alt generativt output **deterministisk**: det samme program producerer altid det samme visuelle.

2.4 Farve (19+ navngivne farver)

```
(ink "red") ; navngivet farve
(ink 255 0 0) ; RGB
(ink 255 0 0 128) ; RGBA (semi-transparent)
(ink "fade:red-blue") ; horisontal gradient
(ink (fade "cyan" "magenta" "vertical"))
(ink "rainbow") ; gentagende regnbue
(ink "zebra") ; stribet mønster
```

2.5 System og display (9 funktioner)

```
(def w width) ; skærmbredde
(def h height) ; skærmhøjde
frame ; nuværende framenummer
time ; sekunder forløbet

(resolution "half") ; halv opløsning
(resolution 64 64) ; eksPLICIT 64x64
(fps 30) ; sæt framerate
```

2.6 Kontrolflow (7 funktioner)

```
; Variabeldefinition
(def x 50)
(def y (+ 10 (* frame 0.5)))
```

```
; Betinget
(if (> frame 60)
  (ink "red")
  (ink "blue"))

; Løkke
(repeat 10 i
  (circle (* i 12) 64 5))

; Eksekver én gang (ikke hvert frame)
(once (wipe "black"))

; Lokal binding
(let ((cx 64) (cy 64))
  (circle cx cy 30))
```

2.7 Timing

KidLisp's timingsystem bruger *inline-planlægning* snarere end callbacks:

```
; Frame-baseret timing
(0 (wipe "black")) ; hvert frame
(1 (spin 5)) ; hvert 2. frame
(30f (ink "red")) ; efter 30 frames

; Sekundbaseret timing
(1s (wipe "red")) ; efter 1 sekund
(2.5s (ink "gold")) ; efter 2,5 sekunder
(1s! (write "!" 10 10)) ; affyr én gang ved 1s

; Gentagende intervaller
(0.5s... (spin 45)) ; hvert 0,5 sekund
(1s... (scroll 1 0)) ; hvert sekund
```

Dette eliminerer behovet for setTimeout, setInterval eller animationsframe-callbacks. Tid er en *egenskab ved udtrykket*, ikke en funktion kaldt på det.

2.8 Tekst (4 funktioner)

```
(write "hello" 10 20) ; tekst ved position
(write "centreret" ; centreret tekst
  (/ width 2) (/ height 2))
(type "typed" 10 10) ; skrivemaskineeffekt
```

2.9 Input (3 funktioner)

```
; Pen/museposition og -tilstand
(if pen
  (circle pen.x pen.y 5))

; Berøringsinput
(if hand
  (line hand.x hand.y 64 64))
```

2.10 Lyd (6 funktioner)

```
; Mikrofoninput
(def vol (mic)) ; amplitude 0-255
```

```
(circle 64 64 vol)      ; lydreaktiv

; Melodiafspilning
(melody "C4 E4 G4 C5")

; Overtoneharmonikker
(overtone 440 3)        ; 3. harmonik af A4
```

2.11 3D-grafik (8 funktioner)

```
(cube "mycube")          ; opret 3D-kube
(form "mycube"
 (move 0 0 -5)           ; position
 (scale 2)               ; størrelse
 (spin (* frame 0.5) 1 1 0)) ; rotér
```

2.12 Komposition via \$-koder

Den mest karakteristiske funktion i KidLisp er dens kompositionssystem. I stedet for brugerdefinerede funktioner refererer programmer til andre programmer via deres korte kode:

```
; Indlejr et andet program som et lag
($cow)                ; eksekvér program "cow"
($nece 10 20)          ; send parametre

; Flerlagskomposition
(wipe "black")
($cow)                 ; baggrundslag
($faim)                ; forgrundslag
```

Programmer hentes fra MongoDB, caches i RAM og IndexedDB og indholdsadresseres via SHA-256. \$-kodesystemet erstatter brugerdefinerede funktioner, moduler og biblioteker med en enkelt social mekanisme: *find et program nogen har lavet, referér det via kode, komponér med det.*

3. EVALUERINGSMODEL

KidLisp reevaluerer sit **hele AST hvert frame**. Der er ingen bevaret tilstand ud over eksplicit definerede variable. **frame-** og **time-**variablerne inkrementeres automatisk. Transformationer som **zoom**, **scroll** og **spin** akkumulerer i framebufferen.

Denne model eliminerer **setup()/draw()-opdelingen** kendt fra Processing [Reas and Fry, 2007]. Enhver linje i et KidLisp-program eksekverer hvert frame.

3.1 Deterministisk tilfældighed

random er seedet fra programmets korte kode. Det samme program producerer identisk output på enhver enhed, hver gang. Dette gør KidLisp-programmer til *re-producerbar generativ kunst*.

3.2 Kaostilstand

Ugyldig input producerer ikke fejlmeddelelser. I stedet detekterer KidLisp lave genkendelsesforhold og renderer kunstnerisk visuelt output. Brugeren ser aldrig "Syntax-Error." De ser farver.

4. DESIGN VIA UDELADELSE

KidLisp mangler bevidst:

- **Brugerdefinerede funktioner:** erstattet af \$-kode komposition
- **Rekursion:** forhindrer uendelige løkker og stakoverløb
- **Fil-I/O:** lærredet er det eneste output
- **Netværk:** programmer kan ikke hente eksterne data
- **Strengmanipulation:** ingen **concat**, **split**, **replace**
- **Fejlmeddelelser:** erstattet af kaostilstand
- **Import/require:** erstattet af \$-koder
- **Objekter/klasser:** data er tal, strenge og lister

Hver udeladelse er en designbeslutning. Fraværet af brugerdefinerede funktioner tvinger komposition gennem det sociale lag. Fraværet af fejlmeddelelser betyder, at begyndere aldrig møder en mur af rød tekst. Fraværet af netværk betyder, at ethvert program er selvstændigt og sikkert at køre.

Dette følger princippet om, at et sprogs begrænsninger definerer dets kreative rum mere end dets funktioner [Pappert, 1980, Compton and Mateas, 2015]. KidLisp er ikke et generelt sprog, der tilfældigvis laver kunst. Det er et kunstsprog, der *ikke kan* gøre andet.

5. KOMPLET INDBYGGET INDEKS

Kategori	Antal	Kernefunktioner
Grafik	9	wipe, ink, line, box, circle
Transformationer	11	scroll, zoom, spin, blur, suck
Matematik	14	+, -, *, /, random, sin, cos
Farve	19+	red, blue, fade, rainbow, zebra
System/Display	9	width, height, frame, time
Kontrolflow	7	def, if, repeat, once, let
Tekst/Output	4	write, type, paste, print
Input	3	pen, hand, gamepad
Lyd	6	mic, melody, overtone, sound
3D-grafik	8	cube, form, trans, move, scale
Komposition	4	\$code, layer, bake, embed
Data	4	list, get, set, let
Total	118	

Table 1: Alle 118 indbyggede funktioner efter kategori.

6. EKSEMPEL: KOMPLET PROGRAM

Et komplet KidLisp-program, der tegner et lydreaktivt partikelfelt med orbital bevægelse:

```

"black"
(def vol (+ 5 (* (mic) 0.2)))
(repeat 50 i
  (def angle (* i 0.1256))
  (def r (+ 20 (* vol (sin (+ (* frame 0.02) angle)
    ))))
  (def cx (+ 64 (* r (cos (+ angle (* frame 0.01))
    ))))
  (def cy (+ 64 (* r (sin (+ angle (* frame 0.01))
    ))))
  (ink (+ 100 (* i 3)) 50 (+ 150 (* i 2)))
  (circle cx cy (+ 1 (random 3))))
(blur 1)
(zoom 1.002)

```

Dette program: sætter en sort baggrund, læser mikrofonamplitude, tegner 50 partikler i orbitale baner moduleret af lyd, farver dem med en beregnet gradient, anvender en let sløring og zoomer langsomt ind. Det kører ved 60 FPS. Det er 12 linjer langt. Det kræver ingen imports, ingen setup-funktion, ingen animationsløkke og ingen hændelseslyttere.

ORCID: 0009-0007-4460-4913

Oversat fra engelsk. Originalversion tilgængelig på <https://papers.aesthetic.computer>

References

Kate Compton and Michael Mateas. Casual creators. In *Proceedings of the International Conference on Computational Creativity*, 2015.

John McCarthy. Recursive functions of symbolic expressions and their computation by machine, part i. *Communications of the ACM*, 3(4):184–195, 1960.

Seymour Papert. *Mindstorms: Children, Computers, and Powerful Ideas*. Basic Books, 1980.

Casey Reas and Ben Fry. *Processing: A Programming Handbook for Visual Designers and Artists*. MIT Press, 2007.