



KidLisp 言語リファレンス



12 カテゴリ 118 個のジェネラティブアート組み込み関数

構文、意味論、タイミング、色彩、 $\$$ -Code 合成システム

@jeffrey

Aesthetic.Computer

ORCID: 0009-0007-4460-4913

<https://aesthetic.computer>

[作業草稿 — 引用不可]

要旨。 KidLisp はジェネラティブアートのための極小 Lisp 方言であり、12 カテゴリにわたる 118 個の組み込み関数を持ち、ユーザー定義関数なし、再帰なし、ファイル I/O なし、ネットワークアクセスなしという設計である。設計上安全であり、KidLisp プログラムはサンドボックスを突破できず、ホストをクラッシュさせることもできず、キャンバス以外のリソースにアクセスすることもできない。本稿は完全な言語リファレンスとして、構文ハイライト付きの例で各組み込み関数を文書化する。タイミングシステム（コールバック不要のフレームベースおよび秒ベースのスケジューリング）、色彩モデル（名前付き色、RGB、グラデーション、パターン）、 $\$$ -code 合成システム（主要な抽象化メカニズムとして参照によりプログラムを埋め込む）、評価モデル（確定的疑似乱数生成器を用いて毎フレーム AST 全体を再評価）を扱う。KidLisp は省略を中心に設計された言語として提示する。何ができないかは何ができると同じく慎重に設計されており、その制約が創作空間を生み出し、59 人の作者がこれまでに 16,634 のプログラムを書いている。

1. 構文

KidLisp は S 式構文を使用する [McCarthy, 1960]。各式はアトム（数値、文字列、シンボル）またはリスト（括弧で囲まれた式）である：

```
; Atoms
42                                ; number
"hello"                          ; string
red                              ; symbol (color name)
```

```
; Lists (function application)
(wipe "navy")           ; clear screen navy
(ink "red")             ; set drawing color
(box 10 10 50 50)       ; draw rectangle
(circle 64 64 30)       ; draw circle
```

演算子なし、中置記法なし、括弧・文字列・数値・シンボル以外の特殊構文なし。算術は前置記法を使用する：

```
(+ 1 2)                ; => 3
(* 3 (+ 1 2))          ; => 9
(- 10 3)               ; => 7
(/ 100 4)              ; => 25
(% 17 5)               ; => 2 (modulo)
```

2. 12 のカテゴリ

2.1 グラフィックス (9 関数)

描画プリミティブは即時モードで動作する——各呼び出しがフレームバッファに直接レンダリングされる。

```
(wipe "black")          ; clear canvas
(ink "cyan")            ; set color
(line 0 0 128 128)      ; diagonal line
(box 10 10 40 40)       ; rectangle
(circle 64 64 20)       ; circle at center
(tri 32 10 10 54 54 54) ; triangle
(plot 64 64)            ; single pixel
(flood 64 64)           ; flood fill
(shape 10 10 50 10 30 50) ; polygon
```

`fill` と `outline` は図形を塗りつぶすか輪郭のみにするかを制御する (Processing 互換)：

```
(fill)                  ; filled shapes
(ink "red")             ; 
(circle 64 64 30)       ; solid red circle

(outline)               ; outlined shapes
(ink "white")           ; 
(circle 64 64 30)       ; white ring
```

2.2 変換 (11 関数)

ピクセルレベルの変換は描画後にフレームバッファに作用する：

```
(scroll 1 0)            ; shift pixels right
(zoom 1.01)             ; slow zoom in
(spin 1)                ; rotate 1 degree
(blur 2)                ; gaussian blur
(contrast 1.2)          ; increase contrast
(suck 0.5)              ; radial displacement
(sort)                  ; sort pixels by brightness
```

変換はフレーム間で累積し、明示的な状態管理なしにアニメーションを生み出す：

```
(wipe "black")
(ink "white")
```

```
(circle 64 64 10)      ; draw circle
(spin 2)                ; rotate 2 deg/frame
(zoom 1.005)            ; grow slightly/frame
; Result: spiraling, growing circle
```

2.3 数学 (14 関数)

```
(sin (* frame 0.05))    ; sine wave
(cos (* frame 0.03))    ; cosine wave
(random)                ; 0-255
(random 10)             ; 0-9
(random 5 15)           ; 5-14
(wiggle 10)             ; random +/-5
(floor 3.7)             ; => 3
(abs -5)                ; => 5
(sqrt 16)               ; => 4
(min 3 7 1)             ; => 1
(max 3 7 1)             ; => 7
```

`random` はプログラムのショートコードで初期化されたシード擬似乱数生成器を使用し、すべての生成出力を**確定的**にする。同じプログラムは常に同じ視覚的結果を生成する。

2.4 色彩 (19 以上の名前付き色)

色は名前、RGB 値、またはグラデーション文字列で指定できる：

```
(ink "red")             ; named color
(ink 255 0 0)           ; RGB
(ink 255 0 0 128)       ; RGBA (semi-transparent)
(ink "fade:red-blue")   ; horizontal gradient
(ink (fade "cyan" "magenta" "vertical"))
(ink "rainbow")         ; repeating rainbow
(ink "zebra")           ; striped pattern
```

利用可能な色名にはすべての CSS 名前付き色が含まれる：red、green、blue、yellow、orange、purple、pink、cyan、magenta、black、white、gray、brown、lime、navy、teal、gold、coral、salmon、crimson、lavender、indigo、violet、turquoise、tomato など。

1 行目に裸の色名を置くと背景色が設定される：

```
"navy"
(ink "gold")
(circle 64 64 30)
; Navy background, gold circle
```

2.5 システムとディスプレイ (9 関数)

```
(def w width)           ; screen width
(def h height)          ; screen height
frame                   ; current frame number
time                   ; seconds elapsed

(resolution "half")     ; half-resolution
(resolution 64 64)      ; explicit 64x64
(fps 30)                ; set frame rate
```

2.6 制御フロー (7 関数)

```
; Variable definition
(def x 50)
(def y (+ 10 (* frame 0.5)))

; Conditional
(if (> frame 60)
  (ink "red")
  (ink "blue"))

; Loop
(repeat 10 i
  (circle (* i 12) 64 5))

; Execute once (not every frame)
(once (wipe "black"))

; Local binding
(let ((cx 64) (cy 64))
  (circle cx cy 30))
```

`repeat` ループはカウンタ変数 (上記の `i`) を 0 から $n-1$ まで束縛する。ネストされた `repeat` は異なる変数名を使用する:

```
(repeat 8 i
  (repeat 8 j
    (box (* i 16) (* j 16) 14 14)))
; 8x8 grid of squares
```

2.7 タイミング

KidLisp のタイミングシステムはコールバックではなくインラインスケジューリングを使用する。タイミング式はコードブロックをラップする:

```
; Frame-based timing
(0 (wipe "black")) ; every frame
(1 (spin 5)) ; every 2nd frame
(30f (ink "red")) ; after 30 frames

; Second-based timing
(1s (wipe "red")) ; after 1 second
(2.5s (ink "gold")) ; after 2.5 seconds
(1s! (write "!" 10 10)) ; fire once at 1s

; Repeating intervals
(0.5s... (spin 45)) ; every 0.5 seconds
(1s... (scroll 1 0)) ; every second
```

これにより `setTimeout`、`setInterval`、アニメーションフレームコールバックの必要性が排除される。時間は式の属性であり、呼び出される関数ではない。

2.8 テキスト (4 関数)

```
(write "hello" 10 20) ; text at position
(write "centered" ; centered text
  (/ width 2) (/ height 2))
(type "typed" 10 10) ; typewriter effect
```

2.9 入力 (3 関数)

```
; Pen/mouse position and state
(if pen
  (circle pen.x pen.y 5))

; Touch input
(if hand
  (line hand.x hand.y 64 64))
```

2.10 オーディオ (6 関数)

```
; Microphone input
(def vol (mic)) ; amplitude 0-255
(circle 64 64 vol) ; sound-reactive

; Melody playback
(melody "C4 E4 G4 C5")

; Overtone harmonics
(overtone 440 3) ; 3rd harmonic of A4
```

2.11 3D グラフィックス (8 関数)

```
(cube "mycube") ; create 3D cube
(form "mycube"
  (move 0 0 -5) ; position
  (scale 2) ; size
  (spin (* frame 0.5) 1 1 0)) ; rotate
```

2.12 \$-Code による合成

KidLisp 最大の特徴はその合成システムである。プログラムはユーザー定義関数を使用せず、ショートコードで他のプログラムを参照する:

```
; Embed another program as a layer
($cow) ; execute program "cow"
($nece 10 20) ; pass parameters

; Multi-layer composition
(wipe "black")
($cow) ; background layer
($faim) ; foreground layer

; Chain multiple programs
(resolution 64 64)
($zobi)
(spin 1)
($grel)
```

プログラムは MongoDB から取得され、RAM と IndexedDB にキャッシュされ、SHA-256 でコンテンツアドレッシングされる。`$-code` システムはユーザー定義関数、モジュール、ライブラリを単一のソーシャルメカニズムに置き換える: 誰かが作ったプログラムを見つけ、コードで参照し、合成する。

最も多く参照されるプログラムは単純な視覚プリミティブであり、構成要素として機能する。このシステムはソーシャル依存グラフを生成する。プログラム A

がプログラム B を埋め込み、プログラム B がプログラム C を埋め込む——観察された使用では最大 7 層の深さに達する。

3. 評価モデル

KidLisp は**毎フレーム AST 全体を再評価する**。明示的に定義された変数以外に保持される状態はない。`frame` と `time` 変数は自動的にインクリメントされる。`zoom`、`scroll`、`spin` などの変換はフレームバッファに累積される。

このモデルは Processing [Reas and Fry, 2007] でよく見られる `setup()`/`draw()` の分離を排除する。KidLisp プログラムの各行は毎フレーム実行される。アニメーションはフレーム依存式から生まれる：

```
(wipe "black")
(ink "white")
; Circle orbits center using sin/cos
(circle
  (+ 64 (* 30 (cos (* frame 0.03))))
  (+ 64 (* 30 (sin (* frame 0.03))))
  8)
```

3.1 確定的ランダム性

`random` のシードはプログラムのショートコードに由来する。同じプログラムは、どのデバイスでも、毎回同じ出力を生成する。これにより KidLisp プログラムは再現可能なジェネラティブアートとなる。コードが作品そのものであり、それを生成するための命令にとどまらない。

3.2 カオスモード

無効な入力——意味不明な文字列、不一致の括弧、ランダムな文字——はエラーメッセージを生成しない。代わりに KidLisp は低認識率（既知語彙の < 30%）と高特殊文字率（> 50%）を検出し、芸術的な視覚出力をレンダリングする。ユーザーは決して「SyntaxError」を見ることがない。色彩が表示される。

4. 省略による設計

KidLisp が意図的に持たないもの：

- **ユーザー定義関数**：\$-code 合成で代替
- **再帰**：無限ループとスタックオーバーフローを防止
- **ファイル I/O**：キャンバスが唯一の出力
- **ネットワーク**：プログラムは外部データを取得できない
- **文字列操作**：concat、split、replace なし
- **エラーメッセージ**：カオスモードで代替
- **Import/require**：\$-code で代替
- **オブジェクト/クラス**：データは数値、文字列、リストのみ

各省略は設計上の決定である。ユーザー定義関数がないことにより、ソーシャルレイヤーを通じた合成が促される。エラーメッセージがないことにより、初心

者は赤いエラーテキストの画面に遭遇しない。ネットワークがないことにより、各プログラムは自己完結的で安全に実行できる。

これは次の原則に従う。言語の制約はその機能と同様に創作空間を定義する [Papert, 1980, Compton and Mateas, 2015]。KidLisp はたまたまアートもできる汎用言語ではない。アート以外のことができない芸術言語である。

5. 組み込み関数の完全索引

カテゴリ	数	主要関数
グラフィックス	9	wipe, ink, line, box, circle
変換	11	scroll, zoom, spin, blur, suck
数学	14	+, -, *, /, random, sin, cos
色彩	19+	red, blue, fade, rainbow, zebra
システム/ディスプレイ	9	width, height, frame, time
制御フロー	7	def, if, repeat, once, let
テキスト/出力	4	write, type, paste, print
入力	3	pen, hand, gamepad
オーディオ	6	mic, melody, overtone, sound
3D グラフィックス	8	cube, form, trans, move, scale
合成	4	\$code, layer, bake, embed
データ	4	list, get, set, let
合計	118	

Table 1: カテゴリ別の全 118 組み込み関数。

6. 例：完全なプログラム

軌道運動する音響反応パーティクルフィールドを描画する完全な KidLisp プログラム：

```
"black"
(def vol (+ 5 (* (mic) 0.2)))
(repeat 50 i
  (def angle (* i 0.1256))
  (def r (+ 20 (* vol (sin (+ (* frame 0.02) angle)
    ))))
  (def cx (+ 64 (* r (cos (+ angle (* frame 0.01)
    ))))
  (def cy (+ 64 (* r (sin (+ angle (* frame 0.01)
    ))))
  (ink (+ 100 (* i 3)) 50 (+ 150 (* i 2)))
  (circle cx cy (+ 1 (random 3))))
(blur 1)
(zoom 1.002)
```

このプログラムは、黒い背景を設定し、マイク振幅を読み取り、音に反応して軌道運動する 50 個のパーティクルを描画し、計算されたグラデーションで着色し、わずかなぼかしを適用し、ゆっくりズームインする。60 FPS で動作する。わずか 12 行。インポート不要、セットアップ関数不要、アニメーションループ不要、イベントリスナー不要。

ORCID: 0009-0007-4460-4913

References

Kate Compton and Michael Mateas. Casual creators. In *Proceedings of the International Conference on Computational Creativity*, 2015.

John McCarthy. Recursive functions of symbolic expressions and their computation by machine, part i. *Communications of the ACM*, 3(4):184–195, 1960.

Seymour Papert. *Mindstorms: Children, Computers, and Powerful Ideas*. Basic Books, 1980.

Casey Reas and Ben Fry. *Processing: A Programming Handbook for Visual Designers and Artists*. MIT Press, 2007.